The Naval Ship Research and Development Center is a U.S. Navy center for laboratory effort directed at achieving improved sea and air vehicles. It was formed in March 1967 by merging the David Taylor Model Basin at Carderock, Maryland and the Marine Engineering Laboratory (now Naval Ship R & D Laboratory) at Annapolis, Maryland. The Mine Defense Laboratory (now Naval Ship R & D Laboratory) Panama City, Florida became part of the Center in November 1967.

Naval Ship Research and Development Center
Washington, D.C. 20034

MAJOR NSRDC ORGANIZATIONAL COMPONENTS



DEPARTMENT OF THE NAVY
NAVAL SHIP RESEARCH AND DEVELOPMENT CENTER
WASHINGTON, D. C. 20034

NUMERICAL CALCULATION OF THE ADDED MASS AND DAMPING COEFFICIENTS OF CYLINDERS OSCILLATING IN OR BELOW A FREE SURFACE

by

J.W. Bedel and C.M. Lee

March 1971

Report 3551

TABLE OF CONTENTS

LIST OF FIGURES

ABSTRACT

The computer program presented herein provides values of the added mass and damping coefficients of infinitely long horizontal cylinders oscillating in or below a free surface. The report includes the theoretical background, the general structure, the details of the input and output schemes, and the program listing.

ADMINISTRATIVE INFORMATION

This study was conducted under the in-house independent research/independent exploratory development program of the Naval Ship Research and Development Center (NSRDC). Funding was provided under Project R01101, Subproject ZR011.01.01.

INTRODUCTION

The main objective of this report is to provide sufficient information to enable the user to properly run computer program YFA4. This program was originally developed by W. Frank and is based on his theory.* It provides the pressures, added mass, and damping of a horizontal cylinder oscillating in heave, sway, or roll while located in or below the free surface.

A brief explanation of the analysis contained in the reference is presented in order to provide the understanding necessary for the most effective usage of the program. The theory deals with a velocity potential problem for an oscillating, horizontal cylinder of infinite length. The

*Frank, W., "Oscillation of Cylinders in or below the Free Surface of Deep Fluids," NSRDC Report 2375 (Oct 1967). Hereafter Report 2375 will be indicated simply as "the reference."

cylinder is symmetric about its vertical centerplane and is either in or below the free surface.* It undergoes forced simple harmonic oscillation of small amplitude in heave, sway, or roll.

The solution to this problem was obtained by the integral-equation method utilizing the Green function which was represented by a pulsating source below the free surface. The sources which satisfy the linearized free-surface condition were distributed along the contour of the cross section of the cylinder. The unknown strengths of the distributed sources were obtained by satisfying the kinematic boundary condition on the cylinder. To solve the integral equation, Frank employed an approximation by replacing the continuous source distribution with a finite number of sources. This approximation amounts to replacing the cylinder contour by a finite number of small straight-line segments along each of which the source density is assumed to be constant. The accuracy of this solution therefore depends on the number of source points on the cylindrical contour.

The program is written in Fortran IV and consists of five subroutines and one function supporting the main program. The program can handle in one run any desired number of cylinders for various depths of submergence and various frequencies of oscillation in heave, sway, or roll. This report provides detailed instructions for preparing the input data to the program and a description of the output generated by the program. The

---

*Hereinafter, "in" or "below" the free surface will be referred to as "floating" or "submerged," respectively.

limitations of the program and some necessary precautions in using it are also given. The user is strongly advised to read the Remarks section carefully before using the program.

TERMINOLOGY

In order to facilitate reader comprehension of the precise meaning of the terms used in the text, the following definitions are given below:

1. Added mass and damping - for bodies in water undergoing oscillations due to an external force $F_o \sin \omega t$ in the x-direction, the equation of motion can be given by:

$$(M + m)\ddot{x} + b\dot{x} + kx = F_o \sin \omega t$$

where M is the body mass,

k is the spring constant (or restoring force coefficient),

x is the displacement of the body from its mean position, and

ω is the radian frequency of oscillation.

As used in this equation, m is called "added mass" and b is called "damping." These quantities are used here for the case of two-dimensional cylindrical cross sections and therefore represent the added mass and damping per unit length of the cylinder.

2. Beam - the beam of a floating cylinder is the width of its cross section at the mean waterline when the cylinder is at its mean position (see Figure 1). The beam of a submerged cylinder is the maximum width of its cross section (see Figure 2).

3. Draft - the draft of a floating cylinder is the vertical distance from the keel to the mean waterline (see Figure 1). The draft of a submerged cylinder is the maximum height of the cross section (see Figure 2).

4. Floating cylinder - this is a cylinder located in the free surface as shown in Figure 1. As far as the computation of hydrodynamic quantities are concerned, the shape of the cylinder above the waterline can be arbitrary.

5. Submerged cylinder - this is a cylinder below the free surface as shown in Figure 2.



Figure 1 - Floating Cylinder



Figure 2 - Submerged Cylinder

PROGRAM DESCRIPTION

The program is written in Fortran IV. The program has four major cycling loops. The order of the cycling process is as shown in Figure 3.



Figure 3 - Order of Cycling Process

The program consists of the following subroutines and functions:

MAIN - This program reads in input data and prints out output results. It serves as a control center to interrelate the supporting subroutines.

SHAPED - This subroutine computes the geometric quantities, such as the length and tangient angle for each segment of the cylinder contour, which are transmitted to Subroutines FIND and FREQ through common statements. The length quantities are nondimensionalized by the half-beam for a floating cylinder and by the half-draft for a submerged cylinder.

SING - This function assigns the positive sign to heave motion and the negative sign to sway and roll motions. This distinction of signs for different modes of motion stems from the symmetric or asymmetric property of the flow about the cylinder.

FIND - This subroutine computes the normal derivatives of the logarithmic singularities on the cylinder contour (see Equations (35) and (42) in the reference).

FREQ - This subroutine computes the source strengths on the contour segments, the pressure in phase with the acceleration of the motion, the pressure in phase with the velocity of the motion, the added mass, and the damping.

DAVID - This subroutine calculates the principal-value integral of the type

$$\text{p.v.} \int_o^\infty \frac{e^{-ik(z-\bar{\zeta})}}{\nu - k} dk$$

as shown in Appendix A of the reference.

MATNIV - This subroutine solves for X in the linear algebraic equation AX = B by means of a matrix inversion.

Figure 4 is a block diagram showing the interconnection between and the process order of the foregoing main program and subprograms.

PROGRAM INPUT

The input setup is shown in Figure 5 and an example is given in Figure 6. The explanation of each card is as follows:

Cards 1-4 - TITO; TITD; TITA; TITV; (12A6 each):

These are titles that are printed out just before the computed output of the program. They may be used to describe the subject investigated by the program (see the section on Program Output). These four cards provide a space for a maximum of 288 letters or characters. Blank cards should be provided if no titles are desired.

Card 5 - NOK, MAXB, NH; (3I6):

NOK is the number of frequencies at which the hydrodynamic quantities will be computed. A maximum of 50 frequencies may be provided. MAXB is



Figure 4 - Program Organization



*More than one card may be needed.

**This setup indicates a case in which only one mode is being calculated.

***This setup indicates a case in which only one Cylinder is being handled.

Figure 5 - Input Setup for YFA4

```
Column No.——►6         12         18      24          36          48          60
Card No.
   ↓
   1      TESTING OF W. FRANK 2-D CYLINDER PROBLEM
   2
   3      EXAMINATION OF ADDED MASS FOR TRIANGULAR CYLINDER
   4
   5           2          9           1
   6          .15                   0.25
   7          .50                   0.0
   8           1         13
   9         0.0                    0.19635    0.39270    0.58905    0.78540
             0.98175                1.17810    1.37445    1.57080    1.37445
             1.17810                0.78540    0.39270    0.0
  10        -2.0                   -1.75      -1.5       -1.25      -1.0
            -0.75                  -0.50      -0.25      -0.000010  -0.000008
            -0.000006              -0.000004  -0.000002  -0.0
  11           1
  12        SUBMERGED TRIANGULAR CYLINDER - HEAVE
  13           0
```

Figure 6 - Listing of Sample Data Cards
(Cards are set up for the output shown
in Figures 9 and 10.)

an integer number which is an argument of the variable XA so that XA (MAXB) corresponds to the half-beam of the cylinder. (See Card 9 for the definition of XA.) The input for MAXB is used only for the purpose of obtaining the half-beam and the sectional area coefficient of the cylinder. These data have no effect on the hydrodynamic quantities. NH is the number of different depths of submergence of the cylinder to be studied. If only a floating cylinder is to be considered, then NH = 1. A maximum of ten depths may be provided.

Card 6 - (CAY(K), K = 1, NOK); (5F12.7):

CAY(K) are the nondimensionalized frequencies to be run and are defined by

$$CAY = \frac{\omega^2 a}{g}$$

where ω is the radian frequency,

a is the half-beam for a floating cylinder,
is the half-draft for a submerged cylinder, and

g is the gravitational acceleration.

The quantities used to determine CAY can be in any consistent dimensional units. The program can handle zero- and infinite-frequency cases by letting the values of CAY be zero and a negative number, respectively. More than one card will be necessary if NOK > 5.

Card 7 - (DEP(J), J = 1, NH), CR; (5F12.7):

DEP(J) are the values of the depths of submergence for the submerged cylinder. (For the floating cylinder, NH=1 and DEP(1) = 0.0). They are measured from the free surface to the top of the submerged cylinder (see Figure 2) in the same scale unit as used for the cylinder dimensions. The data for DEP(J) may require more than one card if NH > 5. CR is the negative vertical distance from the center of rotation to the free surface for a floating cylinder or to the intersection of the vertical line of symmetry and the upper contour of the cylinder for a submerged body. CR is needed only when roll motion is considered. If a roll motion is not considered, the user should provide a zero value for CR.

Card 8 - MD, NON; (2I6):

MD is a control integer whose value depends on whether the cylinder is floating or submerged. MD = 1 for a submerged body and MD = 2 for a floating body. NON is the number of segments to be taken on the right-half of the body contour. To determine the number of segments, see Section III of the Remarks. The maximum number of segments is 45.

Card 9 - (XA(J), J = 1, NUT); (5F12.7):

These are the horizontal offsets of the end points of the line segments located on the cylinder contour. These are measured from the vertical line of symmetry. The scale unit of these data can be arbitrary, provided all other length scales involved for the input data are consistent. The value of NUT is internally defined as NUT = NON+1 where NON, the number of segments on the half-cylinder contour, was read in on Card 8. Since the present program is capable of treating only symmetrical cylinders, the X-coordinates of the points on the body contour are given only for the right-hand side of the cylinder contour. These values should be given as positive numbers. The first input point XA(1) must be that of the intersection of the vertical line of symmetry with the bottom contour of the cylinder and should always be given by XA(1) = 0. The remaining points are read in the counterclockwise direction around the contour. The last point should be the intersecting point of the calm waterline and the cylinder contour for a floating cylinder or the intersection point of the vertical line of symmetry and the upper contour for a submerged body. The foregoing description is shown in Figure 7. Several cards may be needed to provide the input data for XA(J).

Card 10 - (YA(J), J = 1, NUT); (5F12.7):

These are the vertical offsets of the end points of the segments on the cylinder contour. They form ordered pairs of offsets with the XA(J) values. As shown in Figure 7, the YA points for a floating cylinder are measured from the free surface, and the YA points for a submerged body are measured from the intersection of the vertical line of symmetry and the upper contour of the cylinder. Note that for either case they will be entered into the program as negative numbers or zero.

Card 11 - MODE; I(6):

MODE is a control variable whose value depends on the type of oscillation. MODE = 1 for heave, MODE = 2 for sway, and MODE = 3 for roll.

Card 12 - TITLE; (12A6):

The information contained on this card is printed out as a title on the program output. Its primary purpose is to describe the specific geometric shape that the input represents. It may also be used to emphasize the mode that is being considered (see the section on Program Output).

If more than one mode of motion is to be considered, the user should provide additional sets of MODE and TITLE cards just after Card 12. For example, if the user wants to consider sway and roll motions in addition to heave motion, he has to provide four more cards after Card 12. The first extra card should give 2 for MD, the second should describe the mode for sway condition, the third should give 3 for MD, and the fourth should describe the mode for roll condition.



a. Floating Cylinder b. Submerged Cylinder

Figure 7 - Description of Coordinates

Card 13 - MODE; (I6):

This is the entry point for a new cylinder. The variable MODE is used not only for designating the different modes of the oscillation but also for transferring to a new cylinder, if any, after all the desired results have been obtained for the preceding cylinder. This transfer is accomplished by assigning MODE equal to a negative integer to indicate that there is another set of data cards for a new cylinder. The new geometry would be read in starting with Card 5. If MODE = 0, it means that the desired computations have been completed and thus the program should call STOP.

PROGRAM OUTPUT

A typical sample printout for the case of a submerged triangular cylinder is shown in Figures 8 and 9.

Figure 8 shows the input data and geometric characteristics of the cylinder. This output page is printed out each time a new geometry is provided. The first two lines are the printout of the input title data TITO, TITD, TITA, and TITV. (See Input Cards 1-4), where TITD and TITV were inputed as blank cards.

The third and fourth lines are the internally executed printout. Under the heading of INPUT VALUES is the printout of the input values. The variables MD, NON, NOK, CR, CAY, and DEPTHS (DEP) have already been defined in the program input. Below these input values are the coordinates XA(J) and YA(J), where XA and YA are as explained under program input. AREA COEFFICIENT is defined by the cross-sectional area divided by the area of a rectangle with the same beam and draft as the inputed cylinder.

Figure 9 is a typical example of the output of the hydrodynamic quantities. The first line is the printout of the input data of the TITLE card (see Card 12 of program input). This line plus the second line giving the depth ratio H/D (H is the depth of submergence as shown in Figure 2 and D is the half-draft of the body) is printed out for each depth. After these lines come the sets of output data for each frequency of oscillation. The CAY value is the same as described in the program input (see Card 6). NONWL is the nondimensional wave length which represents the ratio of the oscillation generated outgoing wave length to the half-beam (half-draft) for a floating (submerged) cylinder. This value provides the general idea of the magnitude of the oscillation frequency. The nondimensional added mass coefficient AMC is defined as the added mass (or added roll moment of inertia) per unit length of the cylinder divided by ρA. The nondimensional damping coefficient DFC is defined as the damping per unit length of the cylinder divided by $\rho\omega A$. The quantities in the divisor are defined as:

ρ = density of fluid,

ω = radian frequency of oscillation,

$$A = \begin{cases} \frac{\pi}{2}\,(\text{half-beam})^2 & \text{for a floating cylinder in heave or sway.} \\ \pi\,(\text{half-draft})^2 & \text{for a submerged cylinder in heave or sway.} \\ \frac{\pi}{4}\,(\text{half-beam})^4 & \text{for a floating cylinder in roll.} \\ \frac{\pi}{2}\,(\text{half-draft})^4 & \text{for a submerged cylinder in roll} \end{cases}$$

WVH is the ratio of the outgoing wave amplitude to the motion amplitude.

TESTING OF W. FRANK 2-D CYLINDER PROBLEM

EXAMINATION OF ADDED MASS FOR TRIANGULAR CYLINDER

SUBMERGED CYLINDERS OSCILLATING UNDER THE FREE SURFACE

H/D = DEPTH TO TOP OF BODY / HALF DRAFT

INPUT VALUES

MD = 1 NON = 13 NOK = 2 CR = 0.

CAY VALUES

0.1500000 0.2500000

DEPTHS

0.5000000

ABSCISSAS OF CYLINDRICAL CROSS SECTION

0.	0.19635	0.39270	0.58905	0.78540
0.98175	1.17810	1.37445	1.57080	1.37445
1.17810	0.78540	0.39270	0.	

ORDINATES OF CYLINDRICAL CROSS SECTION

-2.00000	-1.75000	-1.50000	-1.25000	-1.00000
-0.75000	-0.50000	-0.25000	-0.00001	-0.00001
-0.00001	-0.00000	-0.00000	0.	

HALF BEAM = 1.57080
DRAFT = 2.00000
AREA COEFFICIENT = 0.50000

Figure 8 - Geometric Output

SUBMERGED TRIANGULAR CYLINDER - HEAVE

HEAVING OSCILLATIONS, H/D = 0.50000

CAY = 0.1500 NONWL = 41.8879

AMC = 5.73358 DFC = 3.53881 WVH = 0.50014

PRESSURES IN PHASE WITH ACCELERATION

0.33350	0.33592	0.32991	0.31565	0.29184
0.25516	0.19750	0.08275	-0.22434	-0.38084
-0.55365	-0.71550	-0.79729		

PRESSURES IN PHASE WITH VELOCITY

-0.24004	-0.25312	-0.26985	-0.29061	-0.31642
-0.34937	-0.39421	-0.47112	-0.64755	-0.73502
-0.83210	-0.92313	-0.96914		

DETERMINANT = 0.85849593E-18

CAY = 0.2500 NONWL = 25.1327

AMC = 2.01477 DFC = 5.19088 WVH = 1.00957

PRESSURES IN PHASE WITH ACCELERATION

0.71930	0.74604	0.76539	0.77850	0.78522
0.78395	0.76987	0.71910	0.52156	0.41059
0.28514	0.16480	0.10291		

PRESSURES IN PHASE WITH VELOCITY

0.05588	0.03895	0.01639	-0.01347	-0.05354
-0.10926	-0.19286	-0.35562	-0.79080	-1.03027
-1.30621	-1.57200	-1.70890		

DETERMINANT = 0.12779346E-17

Figure 9 - Hydrodynamic Output

The pressures in phase with the motion acceleration and velocity shown in the output are the pressures on each segment of the half-contour of the cylinder cross section starting from the bottom segment and proceeding in a counterclockwise direction. The values are the pressures divided by $\rho g h_o$ (where h_o is the amplitude of the oscillation) and are to be read row by row. The value of DETERMINANT is a scaled value of the determinant of the matrix coefficient used in determining the source strengths on the cylinder contour (see Equation (23) of the reference). The significance of the value of DETERMINANT is explained in Section II of REMARKS.

REMARKS

CORRECTIONS TO REPORT 2375

The following changes are corrections to typographical errors and do not affect the numerical results. The corrections are indicated by asterisks.

Page	Equation No.	Correction
8	(26)	$\sum_{i=1}^{N} p_a^{(m)} (x_i, y_i; \omega)\, n_i^{(m)} \lvert S_i^* \rvert$
8	(27)	$\sum_{i=1}^{N} p_v^{(m)} (x_i, y_i; \omega)\, n_i^{(m)} \lvert S_i^* \rvert$
9		$\psi_k = B_k \sin (k\pi x/B) \sinh (k\pi \overset{*}{y}/B)$
14		$\text{P.V.} \int_o^\infty \frac{e^{-ik(z \overset{*}{-} \bar{\zeta})}}{\nu - k} dk$
14	(31)	$\overset{*}{-}\, i\pi\, e^{-i\nu(z-\bar{\zeta})}$

Page	Equation No.	Correction
14	(32)	On second and fourth lines the right side of the equation add: $+\begin{cases} 2\pi i & \text{for } x-\xi>0 \\ 0 & \text{for } x-\xi<0 \end{cases}$
15	(33)	$i\left[\sum_{n=1}^{\infty} \frac{r^n \sin(n\theta)}{n \cdot n!} + \begin{cases} \theta & x-\xi>0 \\ \theta-2\pi & x-\xi<0 \end{cases} \text{for}\right]$
15		$S(r,\theta) = \sum_{n=1}^{\infty} \frac{r^n \sin(n\theta)}{n \cdot n!} + \begin{cases} \theta & x-\xi>0 \\ \theta-2\pi & x-\xi<0 \end{cases} \text{for}$
16	(35)	$\int_{s_j} \left[\frac{1}{2\pi} (\log (z-\zeta)^{*}\right.$ ---
20	(42)	If we rewrite Eq. (42) in the form of

$$\Phi(m) = \frac{1}{2\pi} \sum_{j=1}^{N} Q_j \, R_e\{G_1\} \cos \omega t$$

$$- \sum_{j=1}^{N} Q_{N+j} \, R_e\{G_2\} \sin \omega t,$$ then the

corrected form should be

$$\Phi(m) = \left[\frac{1}{2\pi} \sum_{j=1}^{N} Q_j \, R_e\{G_1\} + \sum_{j=1}^{N} Q_{N+j} \, R_e\{G_2\}\right] \cos \omega t$$

$$+ \left[\frac{1}{2\pi} \sum_{j=1}^{N} Q_{N+j} \, R_e\{G_1\} - \sum_{j=1}^{N} Q_j R_e\{G_2\}\right] \sin \omega t$$

LIMITATIONS OF THE PROGRAM

Irregular Frequencies

As described on page 9 of the reference, there exists for a given floating cylinder a set of discrete frequencies at which the described theory fails to give a correct solution. Such frequencies are called "irregular frequencies." An approximate formula to find these frequencies is given by

$$\omega_j = g\sqrt{\frac{j\pi}{B} \cot h \; (j\pi d/B)} \;, \; j = 1,2 \cdots$$

where g is the gravitational acceleration,

B is the beam of the cylinder, and

d is the draft of the cylinder.

If the value of either DETERMINANT or some other computed value shows sudden discontinuities when plotted versus frequency, the user should regard this as due to the irregular frequency problem and thus should discard all results of these discontinuities.

Unsuitable Cylinder Forms

The present program cannot handle certain cylindrical forms primarily because of the limitation of the built-in function for the arc tangent which is called ATAN2 (X,Y). The range of the angle defined by this function is $-\pi<\theta\leq\pi$. When $X < 0$ and Y approaches zero, the value of the ATAN2 function can èither approach π or $-\pi$ depending on the direction of the approach. This sensitivity of ATAN2 coupled with roundoff error can cause a large error in the computed results of the present program. The following case provides an example:

$$X < 0$$

$$Y = A - \left(\frac{A+A}{2}\right)$$

If A is a floating number, the value of Y could be either $Y \geq 1.E-38$ or $Y \leq -1.E-38$ depending on computer facilities (referring to IBM 7090). The former case ATAN2(X,Y) = π, and the later case ATAN2 (X,Y) = $-\pi$.

In the following two cases, the program may produce wrong results:

1. The cylinder has horizontal lines as part of its top contour (see e.g., Figure 10) and the user's computer yields

$$Y = \left(\frac{A+A}{2}\right) - A \leq -1.E-38$$

where A is a floating number.

2. The cylinder has horizontal lines as part of its bottom contour (see e.g., Figure 11) and the user's computer yields

$$Y = \left(\frac{A+A}{2}\right) - A \geq 1.E-38$$

The following remedies are suggested whenever the user encounters the above two cases:

Case 1 - Replace the horizontal lines on the top contour by the dotted lines as shown in Figure 10 and provide the coordinate data for XA and YA (see Input Cards 9 and 10) based on the modified shape.

Case 2 - Replace the horizontal lines on the bottom contour by the dotted lines as shown in Figure 11 and provide the coordinate data for XA and YA based on the modified shape.

The present program cannot treat cylinders which have two or more points on their half-section boundaries having identical vertical coordinates. Some examples of such cylinders are shown in Figure 12.

DETERMINATION OF NUMBER OF BODY SEGMENTS

There is no definite rule to determine how many segments are required to obtain results that are sufficiently accurate. As more segments are provided, the results become more accurate. A trial-and-error method must be used by the user to determine how many segments will yield the desired accuracy. The use of a large number of segments should be avoided as the time, and thus the cost, increases roughly with the square of the increase in segment numbers.

Nine segments of equal circumferential length have been found to yield satisfactory results for a floating semicircular cylinder. Twelve segments have proved satisfactory for a submerged rectangle. Because the source strengths are averaged over each segment, it should be noted that for extreme cases, such as rectangular bodies, the segment lengths should be smaller near the sharp corners. The user should never use less than six segments for a floating cylinder or less than ten for a submerged cylinder.

ESTIMATED COMPUTER TIME

It is always a difficult task to provide an accurate estimate of the running time of a computer program. The present program is no exception and the difficulty is increased because of independent variables involved. These include the number of segments on the body, frequencies, depths of submergence, modes of motion, and cylinder shapes. Under such



Figure 10 - Alteration for
Upper Horizontal Contour



Figure 11 - Alteration for
Lower Horizontal Contour



Figure 12 - Cylinder Forms Unsuitable for
Program YFA4

circumstances, the best rough estimates that can be provided are based on experience with the IBM 7090 computer at NSRDC. The suggested formula for the IBM 7090 which gives the computer time in minutes for a given cylinder is:

$$0.1\ (A \times B \times C)\left(\frac{N}{12}\right)^2 + 1.5$$

where A is the number of frequencies,

B is the number of depths of submergence,

C is the number of modes of motion, and

N is the number of segments on the half-contour of the cylinder cross section.

ACKNOWLEDGMENTS

The authors acknowledge the significant achievements of the late Mr. W. Frank who developed the program and express their thanks to Mr. V.J. Monacella for his helpful comments and careful review of the manuscript.

APPENDIX

PROGRAM LISTING

```
C     MAIN  PROGRAM  --  YFA4
      DIMENSION XA(46),YA(46)
      COMMON PI,HPI,QPI,TPI,MD,MODE,DPH,CR,RAT,SUR,DEG,JERK,DRT,HBM,SG,N
     1OE,PDM,VOL,DEW,UN,OMEGA,CP,WVH,ID,DOG,IG,SEN(46),CES(46),XX(45),YY
     2(45),DEL(45),SNE(45),CSE(45),FR(45),BLOG(45,45),YLOG(45,45),CON(90
     3,1),CT(90,90),PSI1(45,45),PSI2(45,45),PRA(45),PRV(45),DEP(10)
      COMMON/GR/NOK,NUT,NON,TITLE(12),TITO(12),CAY(50),AMC(50),DFC(50),X
     1(46),Y(46)
      COMMON/FQ/K
      COMMON/TT/TEST
      COMMON/GRPH/TITD(12),DET(50),LAD,LDT,TITA(12),TITV(12),PAK(50),PVK
     1(50),PAS(50),PVS(50),LPV
      COMMON/SHP/MAXB,DUL(45)
      COMMON/MOD/F(5),D(5)
    1 FORMAT(12I6)
    2 FORMAT(5F12.7)
    3 FORMAT(12A6)
    4 FORMAT(////60H      SUBMERGED CYLINDERS OSCILLATING UNDER THE FREE
     1 SURFACE//6X,40H H/D = DEPTH TO TOP OF BODY / HALF DRAFT)
    5 FORMAT(39H0      TRANSFORMATION NOT DEFINED - STOP)
    6 FORMAT(24H0      MATRIX IS SINGULAR)
    7 FORMAT(34H0      HEAVING OSCILLATIONS, H/D = F10.5)
    8 FORMAT(34H0      SWAYING OSCILLATIONS, H/D = F10.5)
    9 FORMAT(33H0      ROLLING OSCILLATIONS ABOUT F10.5,8H  H/D = F10.5)
   10 FORMAT(12H0      CAY = F8.4,13H      NONWL = F8.4)
   11 FORMAT(12H0      AMC = F10.5,9H   DFC = F10.5,9H   WVH = F10.5)
   12 FORMAT(42H0      PRESSURES IN PHASE WITH ACCELERATION//)
   13 FORMAT(5F12.5)
   14 FORMAT(38H0      PRESSURES IN PHASE WITH VELOCITY//)
   29 FORMAT(20H0      DETERMINANT = E15.8)
   34 FORMAT(6X,14HSTRUT DEPTH = F4.2,10H,    ARC = F8.4,8H DEGREES/6X,20
     1HDFC = 0.0 FOR CAY = F4.2)
   35 FORMAT(1H0,12X,12HHALF BEAM = F10.5/17X,8HDRAFT = F10.5/6X,19HAREA
     1 COEFFICIENT = F10.5)
   36 FORMAT(1H0,5X,38HABSCISSAS OF CYLINDRICAL CROSS SECTION//)
   37 FORMAT(1H0,5X,38HORDINATES OF CYLINDRICAL CROSS SECTION//)
   42 FORMAT(1H0,5X,42HIMPULSIVE SURFACE CONDITION, NU VERY LARGE)
   52 FORMAT(////61H      SEMISUBMERGED CYLINDERS OSCILLATING IN THE FRE
     1E SURFACE)
   54 FORMAT(5H MD =,I3,4X,                 6H NON =,I3,4X,6H NOK =,I3,4X
     1,5H CR =,F9.4)
   65 FORMAT(2H1 ,12A6)
   66 FORMAT(///2X,11H CAY VALUES//(5F12.7))
   67 FORMAT(/// 2X,13H INPUT VALUES///)
   68 FORMAT(//2X,7H DEPTHS//(5F12.7))
   69 FORMAT(2H1 ,12A6//)
   70 FORMAT(2X,////)
   71 FORMAT(2H1 )
      PI=3.1415927
      HPI=.5*PI
      QPI=.5*HPI
      TPI=2.*PI
      F(1)=0.52175561
      F(2)=0.39866681
```

```
      F(3)=0.075942450
      F(4)=0.003611758
      F(5)=0.000023369972
      D(1)=0.26356032
      D(2)=1.4134031
      D(3)=3.5964258
      D(4)=7.0858100
      D(5)=12.640801
      READ(5,3)(TITO(J),J=1,12)
      READ(5,3)(TITD(J),J=1,12)
      READ(5,3)(TITA(J),J=1,12)
      READ(5,3)(TITV(J),J=1,12)
      WRITE(6,65) (TITO(J),J=1,12)
      WRITE(6,3)(TITD(J),J=1,12)
      WRITE(6,3)(TITA(J),J=1,12)
      WRITE(6,3)(TITV(J),J=1,12)
   55 READ(5,1) NOK,MAXB,NH
      READ(5,2)(CAY(K),K=1,NOK)
      READ(5,2)(DEP(J),J=1,NH),CR
      READ(5,1) MD,NON
      NUT=NON+1
      READ(5,2) (XA(J),J=1,NUT)
      READ(5,2) (YA(J),J=1,NUT)
      GO TO (61,62),MD
   62 HBEAM=XA(NUT)
      GO TO 63
   61 HBEAM=XA(MAXB)-XA(1)
   63 AREA=0.0
      DO 64 J=1,NON
      XX(J)=.5*(XA(J)+XA(J+1))
      XINT=XA(J+1)-XA(J)
      YINT=YA(J+1)-YA(J)
      DA=YINT*XX(J)
      AREA=AREA+DA
      DUL(J)=SQRT(XINT**2+YINT**2)
      SNE(J)=YINT/DUL(J)
   64 CSE(J)=XINT/DUL(J)
      DRAFT=YA(NUT)-YA(1)
      AREA=AREA/(HBEAM*DRAFT)
      GO TO (50,51),MD
   51 WRITE(6,52)
      GO TO 53
   50 WRITE(6,4)
   53 WRITE(6,67)
      WRITE(6,54) MD,NON,NOK,CR
      WRITE(6,66) (CAY(K),K=1,NOK)
      WRITE(6,68) (DEP(J),J=1,NH)
      WRITE(6,36)
      WRITE(6,13) (XA(J),J=1,NUT)
      WRITE(6,37)
      WRITE(6,13) (YA(J),J=1,NUT)
      WRITE(6,35) HBEAM,DRAFT,AREA
   38 READ(5,1) MODE
      IF(MODE) 40,40,41
   41 READ(5,3)(TITLE(J),J=1,12)
      DO 20 I5=1,NH
```

```
      WRITE(6,69) (TITLE(J),J=1,12)
      DPH=DEP(I5)
      DO 78 J=1,NUT
      X(J)=XA(J)
   78 Y(J)=YA(J)
   17 CALL SHAPED
   18 CALL FIND
   32 GO TO(21,22,23),MODE
   21 WRITE(6,7)DPH
      GO TO 24
   22 WRITE(6,8)DPH
      GO TO 24
   23 WRITE(6,9)CR,DPH
   24 WRITE(6,70)
      DO 20 K=1,NOK
      UN=CAY(K)/PDM
      OMEGA=SQRT(ABS(UN))
      WLN=2.*PI/ABS(CAY(K))
      WN=CAY(K)/DRT
      IF(UN)43,44,44
   43 WRITE(6,42)
      GO TO 45
   44 WRITE(6,10) CAY(K), WLN
   45 CALL FREQ
      ID=ID
      GO TO(25,28),ID
   28 WRITE(6,6)
      GO TO 48
   25 WRITE(6,11)AMC(K),DFC(K),WVH
      WRITE(6,12)
      WRITE(6,13)(PRA(J),J=1,NON)
      WRITE(6,14)
      WRITE(6,13)(PRV(J),J=1,NON)
      WRITE(6,29)DOG
      PAK(K)=PRA(1)
      PVK(K)=PRV(1)
      PAS(K)=PRA(NON)
      PVS(K)=PRV(NON)
      IF(UN)46,46,47
   46 DET(K)=0.0
      GO TO 48
   47 DET(K)=DOG
   48 WRITE(6,70)
   20 CONTINUE
   27 GO TO 38
   40 WRITE(6,71)
      IF(MODE.LT.0) GO TO 55
      STOP
      END
```

```
      SUBROUTINE SHAPED
      COMMON PI,HPI,QPI,TPI,MD,MODE,DPH,CR,RAT,SUR,DEG,JERK,DRT,HBM,SG,N
     1OE,PDM,VOL,DEW,UN,OMEGA,CP,WVH,ID,DOG,IG,SEN(46),CES(46),XX(45),YY
     2(45),DEL(45),SNE(45),CSE(45),FR(45),BLOG(45,45),YLOG(45,45),CON(90
     3,1),CT(90,90),PSI1(45,45),PSI2(45,45),PRA(45),PRV(45)
      COMMON/GR/NOK,NUT,NON,TITLE(12),TITO(12),CAY(50),AMC(50),DFC(50),X
     1(46),Y(46)
      COMMON/SHP/MAXB,DUL(45)
      JERK=1
      KAB=MAXB
      GO TO (10,15),MD
   15 D=X(NUT)
      DPH=0.0
      GO TO 53
   10 D=.5*(Y(NUT)-Y(1))
   53 DO 54 J=1,NUT
      X(J)=X(J)/D
   54 Y(J)=Y(J)/D
      DRT=D
      HBM=X(KAB)
      CP=CR/D
      IF(DPH) 24,24,21
   21 DPH=DPH/D
      DO 22 J=1,NUT
   22 Y(J)=Y(J)-DPH
      NUT=NON+1
   24 DO 25 J=1,NON
      XX(J)=.5*(X(J)+X(J+1))
      YY(J)=.5*(Y(J)+Y(J+1))
   25 DEL(J)=DUL(J)/D
      SG=SING(MODE)
      NOE=2*NON
      GO TO(26,27,28),MODE
   26 DO 29 J=1,NON
   29 FR(J)=CSE(J)
   33 PDM=1.0
      DEW=1.
      GO TO (31,30),MD
   30 VOL=HPI
      GO TO 37
   31 VOL=PI
      GO TO 37
   27 DO 32 J=1,NON
   32 FR(J)=-SNE(J)
      GO TO 33
   28 DO 34 J=1,NON
   34 FR(J)=(YY(J)-CP)*SNE(J)+XX(J)*CSE(J)
      DEW=HBM
      PDM=1.0
      GO TO (36,35),MD
   35 VOL=QPI
      GO TO 37
   36 VOL=HPI
   37 RETURN
      END
```

```
      FUNCTION SING(N)
      IF(1-N)2,1,1
    1 SING=1.
      GO TO 77
    2 SING=-1.
   77 RETURN
      END
```

```
      SUBROUTINE FIND
      COMMON PI,HPI,QPI,TPI,MD,MODE,DPH,CR,RAT,SUR,DEG,JERK,DRT,HBM,SG,N
     1OE,PDM,VOL,DEW,UN,OMEGA,CP,WVH,ID,DOG,IG,SEN(46),CES(46),XX(45),YY
     2(45),DEL(45),SNE(45),CSE(45),FR(45),BLOG(45,45),YLOG(45,45),CON(90
     3,1),CT(90,90),PSI1(45,45),PSI2(45,45),PRA(45),PRV(45)
      COMMON/GR/NOK,NUT,NON,TITLE(12),TITO(12),CAY(50),AMC(50),DFC(50),X
     1(46),Y(46)
      DO 1 I=1,NON
      XM1=XX(I)-X(1)
      YM1=YY(I)-Y(1)
      XP1=XX(I)+X(1)
      YP1=YY(I)+Y(1)
      FPR1=.5*ALOG(XM1**2+YM1**2)
      FPL1=.5*ALOG(XP1**2+YM1**2)
      FCR1=.5*ALOG(XM1**2+YP1**2)
      FCL1=.5*ALOG(XP1**2+YP1**2)
      APR1=ATAN2(YM1,XM1)
      APL1=ATAN2(YM1,XP1)
      ACR1=ATAN2(YP1,XM1)
      ACL1=ATAN2(YP1,XP1)
      DO 1 J=1,NON
      XM2=XX(I)-X(J+1)
      YM2=YY(I)-Y(J+1)
      XP2=XX(I)+X(J+1)
      YP2=YY(I)+Y(J+1)
      FPR2=.5*ALOG(XM2**2+YM2**2)
      FPL2=.5*ALOG(XP2**2+YM2**2)
      FCR2=.5*ALOG(XM2**2+YP2**2)
      FCL2=.5*ALOG(XP2**2+YP2**2)
      APR2=ATAN2(YM2,XM2)
      APL2=ATAN2(YM2,XP2)
      ACR2=ATAN2(YP2,XM2)
      ACL2=ATAN2(YP2,XP2)
      SIMJ=SNE(I)*CSE(J)-SNE(J)*CSE(I)
      CIMJ=CSE(I)*CSE(J)+SNE(I)*SNE(J)
      SIPJ=SNE(I)*CSE(J)+SNE(J)*CSE(I)
      CIPJ=CSE(I)*CSE(J)-SNE(I)*SNE(J)
      DPNR=SIMJ*(FPR1-FPR2)+CIMJ*(APR1-APR2)
   99 PPR=CSE(J)*(XM1*FPR1-YM1*APR1-XM1-XM2*FPR2+YM2*APR2+XM2)+SNE(J)*(Y
     1M1*FPR1+XM1*APR1-YM1-YM2*FPR2-XM2*APR2+YM2)
      DPNL=SIPJ*(FPL2-FPL1)+CIPJ*(APL2-APL1)
      PPL=CSE(J)*(XP2*FPL2-YM2*APL2-XP2-XP1*FPL1+YM1*APL1+XP1)+SNE(J)*(Y
     1M1*FPL1+XP1*APL1+YM2-YM2*FPL2-XP2*APL2-YM1)
      DCNR=SIPJ*(FCR1-FCR2)+CIPJ*(ACR1-ACR2)
      PCR=CSE(J)*(XM1*FCR1-YP1*ACR1-XM1-XM2*FCR2+YP2*ACR2+XM2)+SNE(J)*(Y
     1P2*FCR2+XM2*ACR2+YP1-YP1*FCR1-XM1*ACR1-YP2)
      DCNL=SIMJ*(FCL2-FCL1)+CIMJ*(ACL2-ACL1)
      PCL=CSE(J)*(XP2*FCL2-YP2*ACL2-XP2-XP1*FCL1+YP1*ACL1+XP1)+SNE(J)*(Y
     1P2*FCL2+XP2*ACL2-YP2-YP1*FCL1-XP1*ACL1+YP1)
      BLOG(I,J)=DPNR+SG*DPNL-DCNR-SG*DCNL
      YLOG(I,J)=PPR+SG*PPL-PCR-SG*PCL
      IF(J-NON)2,1,1
    2 XM1=XM2
```

```
      YM1=YM2
      XP1=XP2
      YP1=YP2
      FPR1=FPR2
      FPL1=FPL2
      FCR1=FCR2
      FCL1=FCL2
      APR1=APR2
      APL1=APL2
      ACR1=ACR2
      ACL1=ACL2
    1 CONTINUE
      RETURN
      END
```

```
      SUBROUTINE FREQ
      COMMON PI,HPI,QPI,TPI,MD,MODE,DPH,CR,RAT,SUR,DEG,JERK,DRT,HBM,SG,N
     1OE,PDM,VOL,DEW,UN,OMEGA,CP,WVH,ID,DOG,IG,SEN(46),CES(46),XX(45),YY
     2(45),DEL(45),SNE(45),CSE(45),FR(45),BLOG(45,45),YLOG(45,45),CON(90
     3,1),CT(90,90),PSI1(45,45),PSI2(45,45),PRA(45),PRV(45)
      COMMON/GR/NOK,NUT,NON,TITLE(12),TITO(12),CAY(50),AMC(50),DFC(50),X
     1(46),Y(46)
      COMMON/FQ/K
      COMMON/TT/TEST
      IF(UN)8,9,10
    8 DO 11 I=1,NON
      DO 11 J=1,NON
      CT(I,J)=BLOG(I,J)
   11 PSI1(I,J)=YLOG(I,J)
      GO TO 14
    9 DO 12 I=1,NON
      XM1=XX(I)-X(1)
      XP1=XX(I)+X(1)
      YP1=YY(I)+Y(1)
      FCR1=.5*ALOG(XM1**2+YP1**2)
      FCL1=.5*ALOG(XP1**2+YP1**2)
      ACR1=ATAN2(YP1,XM1)
      ACL1=ATAN2(YP1,XP1)
      DO 12 J=1,NON
      XM2=XX(I)-X(J+1)
      XP2=XX(I)+X(J+1)
      YP2=YY(I)+Y(J+1)
      FCR2=.5*ALOG(XM2**2+YP2**2)
      FCL2=.5*ALOG(XP2**2+YP2**2)
      ACR2=ATAN2(YP2,XM2)
      ACL2=ATAN2(YP2,XP2)
      SIMJ=SNE(I)*CSE(J)-SNE(J)*CSE(I)
      CIMJ=CSE(I)*CSE(J)+SNE(I)*SNE(J)
      SIPJ=SNE(I)*CSE(J)+SNE(J)*CSE(I)
      CIPJ=CSE(I)*CSE(J)-SNE(I)*SNE(J)
      DCNR=SIPJ*(FCR1-FCR2)+CIPJ*(ACR1-ACR2)
      PCR=CSE(J)*(XM1*FCR1-YP1*ACR1-XM1-XM2*FCR2+YP2*ACR2+XM2)+SNE(J)*(Y
     1P2*FCR2+XM2*ACR2+YP1-YP1*FCR1-XM1*ACR1-YP2)
      DCNL=SIMJ*(FCL2-FCL1)+CIMJ*(ACL2-ACL1)
      PCL=CSE(J)*(XP2*FCL2-YP2*ACL2-XP2-XP1*FCL1+YP1*ACL1+XP1)+SNE(J)*(Y
     1P2*FCL2+XP2*ACL2-YP2-YP1*FCL1-XP1*ACL1+YP1)
      CT(I,J)=BLOG(I,J)+2.0*(DCNR+SG*DCNL)
      PSI1(I,J)=YLOG(I,J)+2.0*(PCR+SG*PCL)
      IF(J-NON)13,12,12
   13 XM1=XM2
      XP1=XP2
      YP1=YP2
      FCR1=FCR2
      FCL1=FCL2
      ACR1=ACR2
      ACL1=ACL2
   12 CONTINUE
   14 DO 15 I=1,NON
   15 CON(I,1)=FR(I)
      CALL MATINV(CT,NON,CON,1,DOG,ID)
```

```
      GO TO(16,6),ID
   16 DO 17 I=1,NON
      PRA(I)=0.0
      PRV(I)=0.0
      DO 17 J=1,NON
   17 PRA(I)=PRA(I)-CON(J,1)*PSI1(I,J)
      AMC(K)=0.0
      DFC(K)=0.0
      WVH=0.0
      DO 18 I=1,NON
   18 AMC(K)=AMC(K)+PRA(I)*DEL(I)*FR(I)
      AMC(K)=2.0*AMC(K)/VOL
      GO TO 6
   10 DO 1 I=1,NON
      NI=NON+I
      CON(I,1)=0.
      CON(NI,1)=OMEGA*FR(I)
      XR1=UN*(XX(I)-X(1))
      YR1=-UN*(YY(I)+Y(1))
      XL1=UN*(XX(I)+X(1))
      YL1=YR1
      CALL DAVID(XR1,YR1,EJ1,CXR1,SXR1,RAR1,RBR1,CR1,SR1)
      CALL DAVID(XL1,YL1,EJ1,CXL1,SXL1,RAL1,RBL1,CL1,SL1)
      DO 1 J=1,NON
      NJ=NON+J
      XR2=UN*(XX(I)-X(J+1))
      YR2=-UN*(YY(I)+Y(J+1))
      XL2=UN*(XX(I)+X(J+1))
      YL2=YR2
      CALL DAVID(XR2,YR2,EJ2,CXR2,SXR2,RAR2,RBR2,CR2,SR2)
      CALL DAVID(XL2,YL2,EJ2,CXL2,SXL2,RAL2,RBL2,CL2,SL2)
      SIPJ=SNE(I)*CSE(J)+SNE(J)*CSE(I)
      CIPJ=CSE(I)*CSE(J)-SNE(I)*SNE(J)
      SIMJ=SNE(I)*CSE(J)-SNE(J)*CSE(I)
      CIMJ=CSE(I)*CSE(J)+SNE(I)*SNE(J)
      CT(I,J)=BLOG(I,J)+2.*(SIPJ*(CR1-CR2)-CIPJ*(SR1-SR2)-SG*(SIMJ*(CL1-
     1CL2)-CIMJ*(SL1-SL2)))
      PSI1(I,J)=YLOG(I,J)+2./UN*(SNE(J)*(RAR1-RAR2)+CSE(J)*(RBR1-RBR2)+S
     1G*(SNE(J)*(RAL1-RAL2)+CSE(J)*(RBL2-RBL1)))
      CT(NI,NJ)=CT(I,J)
      CT(I,NJ)=TPI*(EJ2*(SXR2*CIPJ-CXR2*SIPJ)-EJ1*(SXR1*CIPJ-CXR1*SIPJ)-
     1SG*(EJ2*(SXL2*CIMJ-CXL2*SIMJ)-EJ1*(SXL1*CIMJ-CXL1*SIMJ)))
      PSI2(I,J)=TPI/UN*(EJ1*(SXR1*CSE(J)-CXR1*SNE(J))-EJ2*(SXR2*CSE(J)-C
     1XR2*SNE(J))-SG*(EJ1*(SXL1*CSE(J)+CXL1*SNE(J))-EJ2*(SXL2*CSE(J)+CXL
     22*SNE(J))))
      CT(NI,J)=-CT(I,NJ)
      IF(J-NON)7,1,1
    7 XR1=XR2
      YR1=YR2
      XL1=XL2
      YL1=YL2
      EJ1=EJ2
      CR1=CR2
      SR1=SR2
      CL1=CL2
```

```
      SL1=SL2
      RAR1=RAR2
      RBR1=RBR2
      RAL1=RAL2
      RBL1=RBL2
      CXR1=CXR2
      SXR1=SXR2
      CXL1=CXL2
      SXL1=SXL2
    1 CONTINUE
      CALL MATINV(CT,NOE,CON,1,DOG,ID)
      GO TO(2,6),ID
    2 DO 3 I=1,NON
      PRA(I)=0.
      PRV(I)=0.
      DO 4 J=1,NON
      NJ=NON+J
      PRA(I)=PRA(I)+CON(J,1)*PSI2(I,J)-CON(NJ,1)*PSI1(I,J)
    4 PRV(I)=PRV(I)+CON(J,1)*PSI1(I,J)+CON(NJ,1)*PSI2(I,J)
      PRA(I)=OMEGA*PRA(I)
    3 PRV(I)=OMEGA*PRV(I)
      AMC(K)=0.0
      DFC(K)=0.0
      DO 5 I=1,NON
      AMC(K)=AMC(K)+PRA(I)*DEL(I)*FR(I)
    5 DFC(K)=DFC(K)+PRV(I)*DEL(I)*FR(I)
      AMC(K)=2.0*AMC(K)
      DFC(K)=2.0*DFC(K)
      WVH=OMEGA*SQRT(ABS(DFC(K)))/DEW
      AMC(K)=AMC(K)/(UN*VOL)
      DFC(K)=DFC(K)/(UN*VOL)
    6 RETURN
      END
```

```
      SUBROUTINE DAVID(X,Y,E,C,S,RA,RB,CIN,SON)
C     COMPUTATION OF EXPONENTIAL INTEGRAL WITH COMPLEX ARGUMENT
      COMMON/MOD/F(5),D(5)
      Q=3.1415927
      AT=ATAN2(X,Y)
      ARG=AT-.5*Q
      E=EXP(-Y)
      C=COS(X)
      S=SIN(X)
      R=X**2+Y**2
      AL=0.5*ALOG(R)
      A=-Y
      B=-X
      IF(A.GE.0.0) GO TO 78
      IF(B.EQ.0.0) GO TO 79
   78 IF(R.GE.100.) GO TO 10
   79 TEST=0.00001
      IF(R.LT.1.0) GO TO 5
      TEST=0.1*TEST
      IF(R.LT.2.0) GO TO 5
      TEST=0.1*TEST
      IF(R.LT.4.0) GO TO 5
      TEST=0.1*TEST
    5 CONTINUE
      SUMC=0.57721566+AL+Y
      SUMS=AT+X
      TC=Y
      TS=X
      DO 1 K=1,500
      TO=TC
      COX=K
      CAY=K+1
      FACT=COX/CAY**2
      TC=FACT*(Y*TC-X*TS)
      TS=FACT*(Y*TS+X*TO)
      SUMC=SUMC+TC
      SUMS=SUMS+TS
      IF(K.GE.500) GO TO 3
      IF((ABS(TC)+ABS(TS)).GT.TEST) GO TO 1
    3 CIN=E*(C*SUMC+S*SUMS)
      SON=E*(S*SUMC-C*SUMS)
      GO TO 4
    1 CONTINUE
   10 G1=0.
      G2=0.
      DO 20 I=1,5
      DEN=(-Y+D(I))**2+X**2
      GA=F(I)*(-Y+D(I))/DEN
      GB=F(I)*(-X)/DEN
      G1=G1+GA
   20 G2=G2+GB
      CIN=E*Q*S-G1
      SON=-(E*Q*C+G2)
    4 RA=AL-CIN
      RB=ARG+SON
      RETURN
      END
```

```
      SUBROUTINE MATINV(A,N1,B,M1,DETERM,ID)
C      . PIVOT METHOD
C        MATRIX  INVERSION WITH ACCOMPANYING SOLUTION OF SIMUL. EQ.
C     PIVOT METHOD
C     FORTRAN IV SINGLE PRECISION WITH ADJUSTABLE DIMENSION
C     FEBRUARY 1966    S GOOD  DAVID TAYLOR MODEL BASIN    AM MAT4
C          WHERE CALLING PROGRAM MUST INCLUDE
C              DIMENSION A(       ), B(       ), INDEX(      )
C           N      IS THE ORDER OF A
C           M      IS THE NUMBER OF COLUMN VECTORS IN B(MAY BE 0)
C          DETERM WILL CONTAIN DETERMINANT ON EXIT
C           ID     WILL BE SET BY ROUTINE TO 2 IF MATRIX A IS SINGULAR
C                  1 IF INVERSION WAS SUCCESSFUL
C           A      THE INPUT MATRIX WILL BE REPLACED BY A INVERSEE
C           B      THE COLUMN VECTORS WILL BE REPLACED BY CORRESPONDING
C                  SOLUTION VECTORS
C           INDEX WORKING STORAGE ARRAY
C          IF IT IS DESIRED TO SCALE THE DETERMINANT CARD       MAY BE
C          DELETED AND DETERM PRESET BEFORE ENTERING THE ROUTINE
C
      DIMENSION A(90,90),B(90,1),INDEX(90,3)
      EQUIVALENCE (IROW,JROW), (ICOLUM,JCOLUM), (AMAX, T, SWAP)
C      INITIALIZATION
C
C
       N=N1
      M=M1
       DETERM = 1.0
       DO 20 J=1,N
   20 INDEX(J,3) = 0
       DO 550 I=1,N
C
C     SEARCH FOR PIVOT ELEMENT
C
      AMAX = 0.0
      DO 105 J=1,N
      IF(INDEX(J,3)-1) 60, 105, 60
   60 DO 100 K=1,N
      IF(INDEX(K,3)-1) 80, 100, 715
   80 IF (       AMAX -ABS (A(J,K))) 85, 100, 100
   85 IROW=J
      ICOLUM =K
      AMAX = ABS (A(J,K))
  100 CONTINUE
  105 CONTINUE
      INDEX(ICOLUM,3) = INDEX(ICOLUM,3) +1
      INDEX(I,1)=IROW
      INDEX(I,2)=ICOLUM
C
C     INTERCHANGE ROWS TO PUT PIVOT ELEMENT ON DIAGONAL
C
      IF (IROW-ICOLUM) 140, 310, 140
  140 DETERM=-DETERM
      DO 200 L=1,N
      SWAP=A(IROW,L)
```

```
      A(IROW,L)=A(ICOLUM,L)
  200 A(ICOLUM,L)=SWAP
      IF(M) 310, 310, 210
  210 DO 250 L=1, M
      SWAP=B(IROW,L)
      B(IROW,L)=B(ICOLUM,L)
  250 B(ICOLUM,L)=SWAP
C
C     DIVIDE PIVOT ROW BY PIVOT ELEMENT
C
  310 PIVOT    =A(ICOLUM,ICOLUM)
      DETERM=DETERM*PIVOT
  330 A(ICOLUM,ICOLUM)=1.0
      DO 350 L=1,N
  350 A(ICOLUM,L)=A(ICOLUM,L)/PIVOT
      IF(M) 380, 380, 360
  360 DO 370 L=1,M
  370 B(ICOLUM,L)=B(ICOLUM,L)/PIVOT
C
C     REDUCE NON-PIVOT ROWS
C
  380 DO 550 L1=1,N
      IF(L1-ICOLUM) 400, 550, 400
  400 T=A(L1,ICOLUM)
      A(L1,ICOLUM)=0.0
      DO 450 L=1,N
  450 A(L1,L)=A(L1,L)-A(ICOLUM,L)*T
      IF(M) 550, 550, 460
  460 DO 500 L=1,M
  500 B(L1,L)=B(L1,L)-B(ICOLUM,L)*T
  550 CONTINUE
C
C     INTERCHANGE COLUMNS
C
      DO 710 I=1,N
      L=N+1-I
      IF (INDEX(L,1)-INDEX(L,2)) 630, 710, 630
  630 JROW=INDEX(L,1)
      JCOLUM=INDEX(L,2)
      DO 705 K=1,N
      SWAP=A(K,JROW)
      A(K,JROW)=A(K,JCOLUM)
      A(K,JCOLUM)=SWAP
  705 CONTINUE
  710 CONTINUE
      DO 730 K = 1,N
      IF(INDEX(K,3) -1) 715,720,715
 720   CONTINUE
 730   CONTINUE
       ID = 1
 810  RETURN
 715  ID = 2
       GO TO 810
       END
```

INITIAL DISTRIBUTION

Copies	
1	NAVMAT (Code 0331)
9	NAVSEC 1 SEC 6110 1 SEC 6114D 3 SEC 6114H 2 SEC 6136 2 SEC 6111B
5	NAVSHIPS 3 SHIPS 2052 1 SHIPS 034 1 SHIPS 525
3	NAVORDSYSCOM 1 Aero & Hydro Br (Code RAAD-3) 1 Ship Instal & Des (Code SP-26) 1 Loads & Dyn Sec (Code RADD-22)
4	CHONR 1 Nav Analysis (Code 405) 1 Math Br (Code 432) 2 Fluid Dyn (Code 438)
1	ONR, New York
1	ONR, Pasadena
1	ONR, Chicago
1	ONR, Boston
1	ONR, London
1	CDR, USNOL, White Oak
2	DIR, USNRL (Code 5520) 1 Mr. Faires
1	CDR, USNWC, China Lake
1	CDR, NUR&DC, Pasadena
1	CDR, USNAVMISCEN, Point Mugu
1	DIR, Natl BuStand Attn: Dr. Schubauer
12	DDC

Copies	
1	DIR, APL, JHU, Silver Spring
1	DIR, Fluid Mech Lab, Columbia Univ
1	DIR, Fluid Mech Lab, Univ of Calif, Berkeley
1	Prof E.V. Laitone Aero Sci Div Univ of Calif Berkeley, Calif 94720
2	DIR, DL, SIT 1 Dr. H.C. Kim
3	McDonald-Douglas Aircraft, Aircraft Div, Long Beach, Calif 90801 1 Mr. A.M.O. Smith 1 Mr. J.L. Hess 1 Mr. J.P. Giesing
1	DIR, Scripps Inst of Oceano, Univ of Calif
1	DIR, Penn St Univ, Univ Pk
1	DIR, WHOI
1	Prof L. Ward Webb Inst of Nav Arch
2	DIR, Iowa Inst of Hydrau Research 1 Prof L. Landweber
2	DIR, St. Anthony Falls Hydrau Lab 1 Prof C.S. Song
5	Head, NAME, MIT, Cambridge 1 Prof Abkowitz 1 Prof Kerwin 1 Prof Newman 1 Prof Lewis

Copies

2	Inst of Mathematical Sci, NYU, New York 1 Dr. Stoker
1	Hydronautics, Inc., Pindell School Rd Laurel, Md
2	Dept of Nav Architecture, Univ of Calif, Berkeley, Calif 94720 1 Prof J.V. Wehausen 1 Prof W.C. Webster
1	Prof Finn Michelsen, Dept of Nav Arch, Univ of Mich, Ann Arbor
2	Prof T.F. Ogilvie Dept of Nav Arch, Univ of Mich, Ann Arbor
1	Prof Richard MacCamy, Carnegie Tech Pittsburgh 13
2	Hydro Lab, CIT, Pasadena 1 Prof T.Y. Wu
1	Dr. Hartley Pond, 14 Elliot Dr New London, Conn
1	Prof M. Van Dyke, Dept of Aero & Astro Stanford Univ, Palo Alto, Calif
1	Prof B.V. Korvin-Kourkovsky, East Randolph Vt
2	Dept of Aero & Astro, MIT 1 Prof Widnall
1	DIR, Inst for Fluid Dyn & Appl Math Univ of Md
1	DIR, Hydrau Lab, Univ of Colorado
1	Pres, Oceanics, Inc Technical Industrial Park Plainview, N.Y.

UNCLASSIFIED

Security Classification

DOCUMENT CONTROL DATA - R & D

(Security classification of title, body of abstract and indexing annotation must be entered when the overall report is classified)

1. ORIGINATING ACTIVITY *(Corporate author)*

Naval Ship Research and Development Center
Washington, D.C. 20034

2a. REPORT SECURITY CLASSIFICATION

UNCLASSIFIED

2b. GROUP

3. REPORT TITLE

NUMERICAL CALCULATION OF THE ADDED MASS AND DAMPING COEFFICIENTS OF CYLINDERS OSCILLATING IN OR BELOW A FREE SURFACE

4. DESCRIPTIVE NOTES *(Type of report and inclusive dates)*

Final Report

5. AUTHOR(S) *(First name, middle initial, last name)*

J.W. Bedel and C.M. Lee

6. REPORT DATE	7a. TOTAL NO. OF PAGES	7b. NO. OF REFS
March 1971	43	1

8a. CONTRACT OR GRANT NO.

b. PROJECT NO.

R 01101

c. Subproject Task ZR011.01.01

d.

9a. ORIGINATOR'S REPORT NUMBER(S)

3551

9b. OTHER REPORT NO(S) *(Any other numbers that may be assigned this report)*

10. DISTRIBUTION STATEMENT

Approved for public release: distribution unlimited.

11. SUPPLEMENTARY NOTES

12. SPONSORING MILITARY ACTIVITY

NSRDC in-house IR/IED program

13. ABSTRACT

The computer program presented herein provides values of the added mass and damping coefficients of infinitely long horizontal cylinders oscillating in or below a free surface. The report includes the theoretical background, the general structure, the details of the input and output schemes, and the program listing.

DD FORM 1473 1 NOV 65 (PAGE 1)

S/N 0101-807-6801

UNCLASSIFIED

Security Classification

UNCLASSIFIED

Security Classification

14. KEY WORDS	LINK A ROLE	LINK A WT	LINK B ROLE	LINK B WT	LINK C ROLE	LINK C WT
Pressures and Forces on Oscillating Cylinders						
Added Mass Coefficient of Cylinders						
Damping Coefficient of Cylinders						
Two-Dimensional Water Wave Problem						
Numerical Calculations						

DD FORM 1 NOV 65 1473 (BACK)

(PAGE 2)

UNCLASSIFIED

Security Classification